

```

FINISH
/CLEAR
C*****
C***
C*** GRU A BANDIERA
C***
C*** PARAMETRI
C***
/PREP7
H=5000      ! altezza colonna
B=4500      ! sbraccio
HA=3500     ! altezza appoggio braccio verticale
BA=500      ! distanza inserzione braccio verticale
DC=500      ! diametro colonna
SC=5        ! spessore colonna
HB=400      ! altezza profilo braccio orizzontale
BB=400      ! larghezza
SB=5        ! spessore
P=1E5       ! portata
C***
C*** PROPRIETA' GEOMETRICHE SEZIONI
C***
AB=((HB-2*SB)+2*BB)*SB
JB=(SB*(HB-2*SB)**3)/12+2*BB*SB*((HB-SB)/2)**2
AC=3.14*DC*SC
JC=3.14/64*DC**4
C***
C*** NODI
C***
N,1
N,2,0,HA
N,3,DC/2,HA
N,4,0,H
N,5,0,H
N,6,0,H+HB/2
N,7,DC/2,HA
N,8,BA,H+HB/2
N,9,B,H+HB/2
C***
C*** ELEMENTI
C***
ET,1,3
R,1,AC,JC,DC ! colonna
R,2,AB,JB,HB ! braccio
MP,EX,1,210000
C***
C*** colonna
C***
E,1,2
E,2,3
E,2,4
C***
C*** braccio
C***
REAL,2
E,5,6
E,6,8
E,7,8
E,8,9
C***
C*** VINCOLI
C***

```

```

D,1,ALL,0 ! incastro base colonna
CP,1,UX,3,7 ! appoggio orizzontale inferiore colonna-braccio
CP,2,UX,4,5 ! cerniera superiore colonna-braccio
CP,3,UY,4,5
C***
C*** CARICHI
C***
F,9,FY,-P
FINISH
C***
C*** SOLUZIONE
C***
/SOLU
SOLVE
FINISH
C***
C*** POST-PROCESSING
C***
/POST1
SET
C*** opzioni di rappresentazione
/PBC,ALL,,1
/PBC,NFOR,,0
/PBC,NMOM,,0
/PBC,RFOR,,0
/PBC,PATH,,0
/PNUM,NODE,1
C*** scelta disposizione finestra/legenda
/PLOPTS,INFO,2
C*** scelta colori
/COLOR,PBAK,OFF
C*** struttura con vincoli e carichi
/TITLE, MODELLO
EPLLOT
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0
C*** deformata
/PNUM,NODE,0
/TITLE,DEFORMATA
PLDISP,1
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0
C*** caratteristiche di sollecitazione
/PBC,ALL,,0
ETABLE,NI,SMISC,1 ! estrae il dato "sforzo normale" dal data base
ETABLE,NJ,SMISC,7
ETABLE,TI,SMISC,2 ! estrae il dato "taglio" dal data base
ETABLE,TJ,SMISC,8
ETABLE,MI,SMISC,6 ! estrae il dato "momento flettente" dal data base
ETABLE,MJ,SMISC,12
/TITLE,FORZA NORMALE
PLLS,NI,NJ ! rappresenta lo sforzo normale
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0
/TITLE,TAGLIO
PLLS,TI,TJ ! rappresenta il taglio
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0
/TITLE,MOMENTO FLETTENTE
PLLS,MI,MJ ! rappresenta il momento flettente
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

```